
LIX, École Polytechnique, France

INF572 2010/11 – p. 1



The course

Title: Mathematical Programming: Modelling and Software

Code: INF572 (DIX)

Teacher: Leo Liberti (liberti@lix.polytechnique.fr)

Timetable INF572: wed 22,29/9, 6,13,20/10, 3,10,24/11, 15/12

0815-1000 (PC20), 1015-1230 (SI34)

URL INF572: http://www.lix.polytechnique.fr/~liberti/

teaching/inf572-10

INF572 2010/11 – p. 2

http://www.lix.polytechnique.fr/~liberti/
teaching/inf572-10


Contents

1. Introduction

2. AMPL basics

3. AMPL grammar

4. Solvers

5. Mathematical Programming

6. Sudoku

7. Kissing Number Problem

8. Some useful MP theory

9. Reformulations

10. Symmetry

11. The final attack on the KNP

INF572 2010/11 – p. 3



Introduction

Example: Set covering

There are 12 possible geographical positions A1, . . . , A12 where some

discharge water filtering plants can be built. These plants are supposed to

service 5 cities C1, . . . , C5; building a plant at site j (j ∈ {1, . . . , 12}) has

cost cj and filtering capacity (in kg/year) fj ; the total amount of discharge

water produced by all cities is 1.2× 1011 kg/year. A plant built on site j can

serve city i if the corresponding (i, j)-th entry is marked by a ‘*’ in the

table below.

A1 A2 A3 A4 A5 A6 A7 A8 A9 A10 A11 A12

C1 * * * * * *

C2 * * * * * *

C3 * * * * *

C4 * * * * * *

C5 * * * * * * *

cj 7 9 12 3 4 4 5 11 8 6 7 16

fj 15 39 26 31 34 24 51 19 18 36 41 34

What is the best placement for the plants?

INF572 2010/11 – p. 5



Example: Sudoku

Given the Sudoku grid below, find a solution or prove that
no solution exists

2 1

4 1 9 2 8 6

5 8 2 7

5 1 3

9

7 8 6

3 2 6 4 9

1 9 4 5 2 8

8 6

Example: Kissing Number

How many unit balls with disjoint interior can be placed

adjacent to a central unit ball in Rd?

In R2

2 1 0 -1 -2210-1-2

-2

-1

0

1

2

In R3

(D = 3: problem proposed by Newton in 1694, settled by
[Schütte and van der Waerden 1953] and [Leech 1956])

Mathematical programming

The above three problems seemingly have nothing in
common!

Yet, there is a formal language that can be used to
describe all three: mathematical programming (MP)

Moreover, the MP language comes with a rich supply of
solution algorithms so that problems can be solved right
away

Problem

formulation

in MP

→
Reformulation

and choice of so-

lution algorithm

→ Solution process

AMPL → Human intelligence

(for now)
→ Solver

MP language implementations

Software packages implementing (sub/supersets of the) MP language:

AMPL (our software of choice, mixture of MP and near-C language)

commercial, but student version limited to 300 vars/constrs is

available from www.ampl.com

quite a lot of solvers are hooked to AMPL

GNU MathProg (subset of AMPL)

free, but only the GLPK solver (for LPs and MILPs) can be used

it is a significant subset of AMPL but not complete

GAMS (can do everything AMPL can, but looks like COBOL — ugh!)

commercial, limited demo available from www.gams.com

quite a lot of solvers are hooked to GAMS

Zimpl (free, C++ interface, linear modelling only)

LINDO, MPL, . . . (other commercial modelling/solution packages)

How to model
Asking yourself the following questions should help you get started with

your MP model

The given problem is usually a particular instance of a
problem class; you should model the whole class, not just
the instance (replace given numbers by parameter
symbols)

What are the decisions to be taken? Are they logical,
integer or continuous?

What is the objective function? Is it to be minimized or
maximized?

What constraints are there in the problem? Beware —
some constraints may be “hidden” in the problem text

If expressing objective and constraints is overly difficult, go

back and change your variable definitions

Set covering 1

Let us now consider the Set Covering problem

What is the problem class?

We replace the number 12 by the parameter symbol n,

the number 5 by m and the number 1.2× 1011 by d

We already have symbols for costs (cj) and capacities

(fj), where j ≤ n and i ≤ m

We represent the asterisks by a 0-1 matrix A = (aij)

where aij = 1 if there is an asterisk at row i, column j of
the table, and 0 otherwise

Set covering 2
What are the decisions to be taken?

The crucial text in the problem is what is the best placement

for the plants?; i.e. we need to place each plant at some

location

1. geographical placement on a plane? (continuous
variables)

2. yes/no placement? (“should the j-th plant be placed
here?” — logical 0-1 variables)

Because the text also says there are n possible geographical

positions. . . , it means that for each position we have to
decide whether or not to build a plant there

For each of geographical position, introduce a binary

variable (taking 0-1 values):

∀j ≤ n xj ∈ {0, 1}

Set covering 3

What is the objective function?

In this case we only have the indication best placement in
the text

Given our data, two possibilities exist: cost
(minimization) and filtering capacity (maximization)

However, because of the presence of the parameter d, it
wouldn’t make sense to have more aggregated filtering
capacity than d kg/year

Hence, the objective function is the cost, which should
be minimized:

min
∑

j≤n

cjxj

Set covering 4

What are the constraints?

The total filtering capacity must be at least d:

∑

j≤n

fjxj ≥ d

Each city must be served by at least one plant:

∀i ≤ m
∑

j≤n

aijxj ≥ 1

Because there are no more constraints in the text, this
concludes the first modelling phase

Analysis

What category does this mathematical program belong
to?

Linear Programming (LP)

Mixed-Integer Linear Programming (MILP)

Nonlinear Programming (NLP)

Mixed-Integer Nonlinear Programming (MINLP)

Does it have any notable mathematical property?

If an NLP, are the functions/constraints convex?

If a MILP, is the constraint matrix Totally Unimodular
(TUM)?

Does it have any apparent symmetry?

Can it be reformulated to a form for which a fast solver is

available?

Set covering 5

The objective function and all constraints are linear forms

All the decision variables are binary

Hence the problem is a MILP (actually, a BLP)

Good solutions can be obtained via heuristics (e.g. local branching,

feasibility pump, VNS, Tabu Search)

Exact solution via Branch-and-Bound (solver: CPLEX)

No need for reformulation: CPLEX is a fast enough solver

CPLEX 11.0.1 solution: x4 = x7 = x11 = 1, all the rest 0 (i.e. build

plants at positions 4,7,11)

Notice the paradigm model & solver→ solution

Since there are many solvers already available,

solving the problem reduces to modelling the problem

AMPL Basics

AMPL

AMPL means “A Mathematical Programming
Language”

AMPL is an implementation of the Mathematical
Programming language

Many solvers can work with AMPL

AMPL works as follows:

1. translates a user-defined model to a low-level
formulation (called flat form) that can be understood
by a solver

2. passes the flat form to the solver

3. reads a solution back from the solver and interprets
it within the higher-level model (called structured form)

Model, data, run
AMPL usually requires three files:

the model file (extension .mod) holding the MP formulation

the data file (extension .dat), which lists the values to be

assigned to each parameter symbol

the run file (extension .run), which contains the (imperative)

commands necessary to solve the problem

The model file is written in the MP language

The data file simply contains numerical data together with the

corresponding parameter symbols

The run file is written in an imperative C-like language (many notable

differences from C, however)

Sometimes, MP language and imperative language commands can

be mixed in the same file (usually the run file)

To run AMPL, type ampl < problem.run from the command line

INF572 2010/11 – p. 19



An elementary run file

Consider the set covering problem, suppose we have
coded the model file (setcovering.mod) and the data
file (setcovering.dat), and that the CPLEX solver is
installed on the system

Then the following is a basic setcovering.run file

# basic run file for setcovering problem

model setcovering.mod; # specify model file

data setcovering.dat; # specify data file

option solver cplex; # specify solver

solve; # solve the problem

display cost; # display opt. cost

display x; # display opt. soln.

Set covering model file

# setcovering.mod

param m integer, >= 0;

param n integer, >= 0;

set M := 1..m;

set N := 1..n;

param c{N} >= 0;

param a{M,N} binary;

param f{N} >= 0;

param d >= 0;

var x{j in N} binary;

minimize cost: sum{j in N} c[j]*x[j];

subject to capacity: sum{j in N} f[j]*x[j] >= d;

subject to covering{i in M}: sum{j in N} a[i,j]*x[j] >= 1;

Set covering data file
param m := 5;

param n := 12;

param : c f :=

1 7 15

2 9 39

3 12 26

4 3 31

5 4 34

6 4 24

7 5 51

8 11 19

9 8 18

10 6 36

11 7 41

12 16 34 ;

param a: 1 2 3 4 5 6 7 8 9 10 11 12 :=

1 1 0 1 0 1 0 1 1 0 0 0 0

2 0 1 1 0 0 1 0 0 1 0 1 1

3 1 1 0 0 0 1 1 0 0 1 0 0

4 0 1 0 1 0 0 1 1 0 1 0 1

5 0 0 0 1 1 1 0 0 1 1 1 1 ;

AMPL+CPLEX solution
liberti@nox$ cat setcovering.run | ampl

ILOG CPLEX 11.010, options: e m b q use=2

CPLEX 11.0.1: optimal integer solution; objective 15

3 MIP simplex iterations

0 branch-and-bound nodes

cost = 15

x [*] :=

1 0

2 0

3 0

4 1

5 0

6 0

7 1

8 0

9 0

10 0

11 1

12 0

;

AMPL Grammar

AMPL MP Language
There are 5 main entities: sets, parameters, variables, objectives and

constraints

In AMPL, each entity has a name and can be quantified

set name [{quantifier}] attributes ;
param name [{quantifier}] attributes ;
var name [{quantifier}] attributes ;
minimize | maximize name [{quantifier}]: iexpr ;

subject to name [{quantifier}]: iexpr <= | = | >= iexpr ;

Attributes on sets and parameters is used to validate values read

from data files

Attributes on vars specify integrality (binary, integer) and limit

constraints (>= lower, <= upper)

Entities indices: square brackets (e.g. y[1], x[i,k])

The above is the basic syntax — there are some advanced options

INF572 2010/11 – p. 25



AMPL data specification

In general, syntax is in map-like form; a

param p{i in S} integer;

is a map S → Z, and each pair (domain, codomain) must be
specified:

param p :=

1 4

2 -3

3 0;

The grammar is simple but tedious, best way is
learning by example or trial and error

AMPL imperative language
model model filename.mod ;

data data filename.dat ;

option option name literal string, ... ;

solve ;

display [{quantifier}] iexpr ; / printf (syntax similar to C)

let [{quantifier}] ivar :=number;

if (condition list) then { commands } [else {commands}]
for {quantifier} {commands} / break; / continue;

shell ’command line’; / exit number; / quit;

cd dir name; / remove file name;

In all output commands, screen output can be redirected to a file by

appending > output filename.txt before the semicolon

These are basic commands, there are some advanced ones

Reformulation commands

fix [{quantifier}] ivar [:=number];

unfix [{quantifier}] ivar;
delete entity name;

purge entity name;

redeclare entity declaration;

drop/restore [{quantifier}] constr or obj name;

problem name[{quantifier}] [:entity name list] ;

This list is not exhaustive

Solvers

Solvers
In order of solver reliability / effectiveness:

1. LPs: use an LP solver (O(106) vars/constrs, fast, e.g. CPLEX, CLP,

GLPK)

2. MILPs: use a MILP solver (O(104) vars/constrs, can be slow,

e.g. CPLEX, Symphony, GLPK)

3. NLPs: use a local NLP solver to get a local optimum (O(104)

vars/constrs, quite fast, e.g. SNOPT, MINOS, IPOPT)

4. NLPs/MINLPs: use a heuristic solver to get a good local optimum

(O(103), quite fast, e.g. BONMIN, MINLP_BB)

5. NLPs: use a global NLP solver to get an (approximated) global

optimum (O(103) vars/constrs, can be slow, e.g. COUENNE, BARON)

6. MINLPs: use a global MINLP solver to get an (approximated) global

optimum (O(103) vars/constrs, can be slow, e.g. COUENNE, BARON)

Not all these solvers are available via AMPL

Solution algorithms (linear)

LPs: (convex)

1. simplex algorithm (non-polynomial complexity but
very fast in practice, reliable)

2. interior point algorithms (polynomial complexity,
quite fast, fairly reliable)

MILPs: (nonconvex because of integrality)

1. Local (heuristics): Local Branching, Feasibility Pump
[Fischetti&Lodi 05], VNS [Hansen et al. 06] (quite
fast, reliable)

2. Global: Branch-and-Bound (exact algorithm,
non-polynomial complexity but often quite fast,
heuristic if early termination, reliable)

Solution algorithms (nonlinear)

NLPs: (may be convex or nonconvex)

1. Local: Sequential Linear Programming (SLP), Sequential

Quadratic Programming (SQP), interior point methods

(linear/polynomial convergence, often quite fast, unreliable)

2. Global: spatial Branch-and-Bound [Smith&Pantelides 99]

(ε-approximate, nonpolynomial complexity, often quite slow,

heuristic if early termination, unreliable)

MINLPs: (nonconvex because of integrality and terms)

1. Local (heuristics): Branching explorations [Fletcher&Leyffer 99],

Outer approximation [Grossmann 86], Feasibility pump [Bonami

et al. 06] (nonpolynomial complexity, often quite fast, unreliable)

2. Global: spatial Branch-and-Bound [Sahinidis&Tawarmalani 05]

(ε-approximate, nonpolynomial complexity, often quite slow,

heuristic if early termination, unreliable)

Canonical MP formulation

minx f(x)

s.t. l ≤ g(x) ≤ u

xL ≤ x ≤ xU

∀i ∈ Z ⊆ {1, . . . , n} xi ∈ Z







[P ] (1)

where x, xL, xU ∈ Rn; l, u ∈ Rm; f : Rn → R; g : Rn → Rm

A point x∗ is feasible in P if l ≤ g(x∗) ≤ u, xL ≤ x∗ ≤ xU and

∀i ∈ Z (x∗
i ∈ Z); F (P ) = set of feasible points of P

If gi(x
∗) = l or = u for some i, gi is active at x∗

A feasible x∗ is a local minimum if ∃B(x∗, ε) s.t. ∀x ∈ F (P ) ∩B(x∗, ε)

we have f(x∗) ≤ f(x)

A feasible x∗ is a global minimum if ∀x ∈ F (P ) we have f(x∗) ≤ f(x)

Feasibility and optimality

F (P ) = feasible region of P , L(P ) = set of local optima,
G(P ) = set of global optima

Nonconvexity⇒ G(P ) ( L(P )

min
x∈[−3,6]

1
4x + sin(x) −3

60 x

Convexity
A function f(x) is convex if the following holds:

∀x0, x1 ∈ dom(f) ∀λ ∈ [0, 1]

f(λx0 + (1− λ)x1) ≤ λf(x0) + (1− λ)f(x1)

x

f

x0 x1

f(x0)

f(x1)

λx0 + (1 − λ)x1

f(λx0 + (1 − λ)x1)

λf(x0) + (1 − λ)f(x1)

A set C ⊆ Rn is convex if ∀x0, x1 ∈ C, λ ∈ [0, 1] (λx0 + (1− λ)x1 ∈ C)

If g : Rm → Rn are convex, then {x | g(x) ≤ 0} is a convex set

If f, g are convex, then every local optimum of min
g(x)≤0

f(x) is global

A local NLP solver suffices to solve the NLP to optimality

Canonical form

P is a linear programming problem (LP) if f : Rn → R,
g : Rn → Rm are linear forms

LP in canonical form:

minx cTx

s.t. Ax ≤ b

x ≥ 0







[C] (2)

Can reformulate inequalities to equations by adding a
non-negative slack variable xn+1 ≥ 0:

n∑

j=1

ajxj ≤ b ⇒
n∑

j=1

ajxj + xn+1 = b ∧ xn+1 ≥ 0

Standard form

LP in standard form: all inequalities transformed to
equations

minx (c′)
T
x

s.t. A′x = b

x ≥ 0







[S] (3)

where x = (x1, . . . , xn, xn+1, . . . , xn+m),
A′ = (A, Im), c′ = (c, 0, . . . , 0

︸ ︷︷ ︸

m

)

Standard form useful because linear systems of
equations are computationally easier to deal with than
systems of inequalities

Used in simplex algorithm

Geometry of LP

A polyhedron is the intersection of a finite number of
closed halfspaces. A bounded, non-empty polyhedron
is a polytope

x1

x2

x1 + 2x2 ≤ 2

2x1 + x2 ≤ 2

P

Q

R

S

row 1

row 2

Canonical feas. polyhedron: F (C) =
{x ∈ Rn | Ax ≤ b ∧ x ≥ 0}
A =

0

@

1 2

2 1

1

A, bT = (2, 2)

Standard feas. polyhedron: F (S) =

{(x, y) ∈ Rn+m | Ax + Imy =
b ∧ (x, y) ≥ 0}

P = (0, 0, 2, 2), Q = (0, 1, 0, 1), R = (2
3 , 2

3 , 0, 0), S = (1, 0, 1, 0)

Each vertex corresponds to an intersection of at least n
hyperplanes⇒ ≥ n coordinates are zero

Basic feasible solutions

Consider polyhedron in “equation form”
K = {x ∈ Rn | Ax = b ∧ x ≥ 0}. A is m× n of rank m
(N.B. n here is like n + m in last slide!)

A subset of m linearly independent columns of A is a
basis of A

If β is the set of column indices of a basis of A,
variables xi are basic for i ∈ β and nonbasic for i 6∈ β

Partition A in a square m×m nonsingular matrix B
(columns indexed by β) and an (n−m)×m matrix N

Write A = (B|N), xB ∈ Rm basics, xN ∈ Rn−m nonbasics

Given a basis (B|N) of A, the vector x = (xB, xN ) is a
basic feasible solution (bfs) of K with respect to the given
basis if Ax = b, xB ≥ 0 and xN = 0

Fundamental Theorem of LP

Given a non-empty polyhedron K in “equation form”,
there is a surjective mapping between bfs and vertices
of K

For any c ∈ Rn, either there is at least one bfs that

solves the LP min{cTx |x ∈ K}, or the problem is
unbounded

Proofs not difficult but long (see lecture notes or
Papadimitriou and Steiglitz)

Important correspondence between bfs’s and vertices
suggests geometric solution method based on exploring
vertices of K

INF572 2010/11 – p. 40



Simplex Algorithm: Summary

Solves LPs in form min
x∈K

cTx where K = {Ax = b ∧ x ≥ 0}

Starts from any vertex x

Moves to an adjacent improving vertex x′

(i.e. x′ is s.t. ∃ edge {x, x′} in K and cTx′ ≤ cTx)

Two bfs’s with basic vars indexed by sets β, β′

correspond to adjacent vertices if |β ∩ β′| = m− 1

Stops when no such x′ exists

Detects unboundedness and prevents cycling⇒
convergence

K convex⇒ global optimality follows from local
optimality at termination

Simplex Algorithm I

Let x = (x1, . . . , xn) be the current bfs, write Ax = b as
BxB + NxN = b

Express basics in terms of nonbasics:

xB = B−1b−B−1NxN (this system is a dictionary)

Express objective function in terms of nonbasics:

cTx = cT
BxB + cT

NxN = cT
B(B−1b−B−1NxN ) + cT

NxN ⇒
⇒ cTx = cT

BB−1b + c̄T
NxN

(c̄T
N = cT

N − cT
BB−1N are the reduced costs)

Select an improving direction: choose a nonbasic
variable xh with negative reduced cost; increasing its
value will decrease the objective function value

If no such h exists, no improving direction, local
minimum⇒ global minimum⇒ termination

Simplex Algorithm II

Iteration start: xh is out of basis⇒ its value is zero

We want to increase its value to strictly positive to
decrease objective function value

. . . corresponds to “moving along an edge”

We stop when we reach another (improving) vertex

. . . corresponds to setting a basic variable xl to zero

row 2

x1

x2

P

Q

R: optimum

S

row 1

S = (1, 0, 1, 0)

x1 enters, x4 exits

xh enters the basis, xl exits the basis

Simplex Algorithm III

How do we determine l and new positive value for xh?

Recall dictionary xB = B−1b−B−1NxN ,

write b̄ = B−1b and Ā = (āij) = B−1N

For i ∈ β (basics), xi = b̄i −
∑

j 6∈β āijxj

Consider nonbasic index h of variable entering basis (all

the other nonbasics stay at 0), get xi = b̄i − āihxh,∀i ∈ β

Increasing xh may make xi < 0 (infeasible), to prevent

this enforce ∀i ∈ β (b̄i − āihxh ≥ 0)

Require xh ≤ b̄i

āih
for i ∈ β and āih > 0:

l = argmin{ b̄i

āih
| i ∈ β ∧ āih > 0}, xh =

b̄l

ālh

If all āih ≤ 0, xh can increase without limits: problem
unbounded

Simplex Algorithm IV

Suppose > n hyperplanes cross at vtx R (degenerate)

May get improving direction s.t. adjacent vertex is still R

Objective function value does not change

Seq. of improving dirs. may fail to move away from R

⇒ simplex algorithm cycles indefinitely

Use Bland’s rule: among candidate entering / exiting
variables, choose that with least index

x1

x2

x1 + 2x2 ≤ 2

2x1 + x2 ≤ 2

3x1 + 3x2 ≤ 4

P

Q

R: crossing of three lines

S

Example: Formulation

Consider problem:

max
x1,x2

x1 + x2

s.t. x1 + 2x2 ≤ 2

2x1 + x2 ≤ 2

x ≥ 0







Standard form:

−minx −x1 − x2

s.t. x1 + 2x2 + x3 = 2

2x1 + x2 + x4 = 2

x ≥ 0







Obj. fun.: max f = −min−f , simply solve for min−f

Example, itn 1: start

Objective function vector cT = (−1,−1, 0, 0)

Constraints in matrix form:

(

1 2 1 0

2 1 0 1

)








x1

x2

x3

x4








=

(

2

2

)

Choose obvious starting basis with

B =

(

1 0

0 1

)

, N =

(

1 2

2 1

)

, β = {3, 4}

Corresponds to point P = (0, 0, 2, 2)

Example, itn 1: dictionary

Start the simplex algorithm with basis in P

row 1

row 2

x1

x2

x1 + 2x2 ≤ 2

2x1 + x2 ≤ 2

P

Q

R

S

−∇f

Compute dictionary xB = B−1b−B−1NxN = b̄− ĀxN ,
where

b̄ =

(

2

2

)

; Ā =

(

1 2

2 1

)

Example, itn 1: entering var

Compute reduced costs c̄N = cT
N − cT

BĀ:

(c̄1, c̄2) = (−1,−1)− (0, 0)Ā = (−1,−1)

All nonbasic variables {x1, x2} have negative reduced
cost, can choose whichever to enter the basis

Bland’s rule: choose entering nonbasic with least index

in {x1, x2}, i.e. pick h = 1 (move along edge PS)

row 1

row 2

x1

x2

x1 + 2x2 ≤ 2

2x1 + x2 ≤ 2

P

Q

R

S

−∇f

x1 enters the basis

Example, itn 1: exiting var

Select exiting basic index l

l = argmin{ b̄i

āih
| i ∈ β ∧ āih > 0} = argmin{ b̄1

ā11
,

b̄2

ā21
}

= argmin{2
1
,
2

2
} = argmin{2, 1} = 2

Means: “select second basic variable to exit the basis”,
i.e. x4

Select new value b̄l

ālh
for xh (recall h = 1 corrresponds to

x1):

b̄l

ālh
=

b̄2

ā21
=

2

2
= 1

x1 enters, x4 exits (apply swap (1, 4) to β)

Example, itn 2: start

Start of new iteration: basis is β = {1, 3}

B =

(

1 1

2 0

)

; B−1 =

(

0 1
2

1 −1
2

)

xB = (x1, x3) = B−1b = (1, 1), thus current bfs is
(1, 0, 1, 0) = S

�
�
�
�������������

������������
������������
������������
������������
������������
������������
������������
������������
������������
������������

������������
������������
������������
������������
������������
������������
������������
������������
������������
������������
������������

����������������
����������������
����������������
����������������
����������������
����������������
����������������
����������������
����������������
����������������

����������������
����������������
����������������
����������������
����������������
����������������
����������������
����������������
����������������
����������������

������
������
������
������
������
������
������
������
������
������
������
������

������
������
������
������
������
������
������
������
������
������
������
������

row 1

row 2

x1

x2

x1 + 2x2 ≤ 2

2x1 + x2 ≤ 2

P

Q

R

S

−∇f

Example, itn 2: entering var

Compute dictionary: b̄ = B−1b = (1, 1)T,

Ā = B−1N =

(

0 1
2

1 −1
2

)(

2 0

1 1

)

=

(
1
2

1
2

3
2 −1

2

)

Compute reduced costs:
(c̄2, c̄4) = (−1, 0)− (−1, 0)Ā = (−1/2, 1/2)

Pick h = 1 (corresponds to x2 entering the basis)

row 1

x1

x2

x1 + 2x2 ≤ 2

2x1 + x2 ≤ 2

P

Q

R

S

x2 enters basis

−∇f

Example, itn 2: exiting var

Compute l and new value for x2:

l = argmin{ b̄1

ā11
,

b̄2

ā21
} = argmin{ 1

1/2
,

1

3/2
} =

= argmin{2, 2/3} = 2

l = 2 corresponds to second basic variable x3

New value for x2 entering basis: 2
3

x2 enters, x3 exits (apply swap (2, 3) to β)

Example, itn 3: start

Start of new iteration: basis is β = {1, 2}

B =

(

1 2

2 1

)

; B−1 =

(

−1
3

2
3

2
3 −1

3

)

xB = (x1, x2) = B−1b = (2
3 , 2

3), thus current bfs is

(2
3 , 2

3 , 0, 0) = R

row 1

row 2

x1

x2

x1 + 2x2 ≤ 2

2x1 + x2 ≤ 2

P

Q

R

S

−∇f

Example, itn 3: termination

Compute dictionary: b̄ = B−1b = (2/3, 2/3)T,

Ā = B−1N =

(

−1
3

2
3

2
3 −1

3

)(

1 0

0 1

)

=

(

−1
3

2
3

2
3 −1

3

)

Compute reduced costs:
(c̄3, c̄4) = (0, 0)− (−1,−1)Ā = (1/3, 1/3)

No negative reduced cost: algorithm terminates

Optimal basis: {1, 2}
Optimal solution: R = (2

3 , 2
3)

Optimal objective function value f(R) = −4
3

Permutation to apply to initial basis {3, 4}: (1, 4)(2, 3)

Interior point methods

Simplex algorithm is practically efficient but nobody ever
found a pivot choice rule that proves that it has
polynomial worst-case running time

Nobody ever managed to prove that such a rule does
not exist

With current pivoting rules, simplex worst-case running
time is exponential

Kachiyan managed to prove in 1979 that LP ∈ P using a
polynomial algorithm called ellipsoid method

(http://www.stanford.edu/class/msande310/ellip.pdf)

Ellipsoid method has polynomial worst-case running
time but performs badly in practice

Barrier interior point methods for LP have both
polynomial running time and good practical
performances

IPM I: Preliminaries

Consider LP P in standard form:
min{cTx | Ax = b ∧ x ≥ 0}.
Reformulate by introducing “logarithmic barriers”:

P (β) : min{cTx− β

n∑

j=1

log xj | Ax = b}

−β log(x)

x

decreasing β

The term −β log(xj) is a

“penalty” that ensures that
xj > 0; the “limit” of this

reformulation for β → 0 should
ensure that xj ≥ 0 as desired

Notice P (β) is convex ∀β > 0

IPM II: Central path

Let x∗(β) the optimal solution of P (β) and x∗ the optimal
solution of P

The set {x∗(β) | β > 0} is called the central path

Idea: determine the central path by
solving a sequence of convex
problems P (β) for some decreasing
sequence of values of β and show
that x∗(β)→ x∗ as β → 0

Since for β > 0, −β log(xj) → +∞
for xj → 0, x∗(β) will never be on the

boundary of the feasible polyhedron
{x ≥ 0 | Ax = b}; thus the name “inte-
rior point method”

0 0.5 1 1.5 2 2.5 3
0

0.4

0.6

0.8

1

1.2

2

Optimality Conditions I

If we can project improving direction −∇f(x′) on an
active constraint g2 and obtain a feasible direction d,
point x′ is not optimal

x1

x2

x′g1

g2

∇g1(x′)

∇g1(x′)

−∇f(x′)

C

d

Implies −∇f(x′) 6∈ C (cone generated by active constraint

gradients)

Optimality Conditions II

Geometric intuition: situation as above does not happen
when −∇f(x∗) ∈ C, x∗ optimum

x1

x∗

g1

g2

∇g1(x∗)

∇g1(x∗)

−∇f(x∗)

C

Projection of −∇f(x∗) on active constraints is never a
feasible direction

Optimality Conditions III

If:

1. x∗ is a local minimum of problem
P ≡ min{f(x) | g(x) ≤ 0},

2. I is the index set of the active constraints at x∗,
i.e. ∀i ∈ I (gi(x

∗) = 0)

3. ∇gI(x
∗) = {∇gi(x

∗) | i ∈ I} is a linearly independent
set of vectors

then −∇f(x∗) is a conic combination of ∇gI(x
∗),

i.e. ∃y ∈ R|I| such that

∇f(x∗) +
∑

i∈I

yi∇gi(x
∗) = 0

∀i ∈ I yi ≥ 0

Karush-Kuhn-Tucker Conditions

Define

L(x, y) = f(x) +
m∑

i=1

yigi(x)

as the Lagrangian of problem P

KKT: If x∗ is a local minimum of problem P and ∇g(x∗)
is a linearly independent set of vectors, ∃y ∈ Rm s.t.

∇x∗L(x, y) = 0

∀i ≤ m (yigi(x
∗) = 0)

∀i ≤ m (yi ≥ 0)

INF572 2010/11 – p. 62



Weak duality

Thm.

Let L̄(y) = min
x∈F (P )

L(x, y) and x∗ be the global optimum

of P . Then ∀y ≥ 0 L̄(y) ≤ f(x∗).

Proof

Since y ≥ 0, if x ∈ F (P ) then yigi(x) ≤ 0, hence
L(x, y) ≤ f(x); result follows as we are taking the mini-
mum over all x ∈ F (P ).

Important point: L̄(y) is a lower bound for P for all y ≥ 0

The problem of finding the tightest Lagrangian lower
bound

max
y≥0

min
x∈F (P )

L(x, y)

is the Lagrangian dual of problem P

Dual of an LP I

Consider LP P in form: min{cTx | Ax ≥ b ∧ x ≥ 0}
L(x, s, y) = cTx− sTx + yT(b− Ax) where s ∈ Rn, y ∈ Rm

Lagrangian dual:

max
s,y≥0

min
x∈F (P )

(yb + (cT − s− yA)x)

KKT: for a point x to be optimal,

cT − s− yA = 0 (KKT1)

∀j ≤ n (sjxj = 0), ∀i ≤ m (yi(bi − Aix) = 0) (KKT2)

s, y ≥ 0 (KKT3)

Consider Lagrangian dual s.t. (KKT1), (KKT3):

Dual of an LP II

Obtain:

max
s,y

yb

s.t. yA + s = cT

s, y ≥ 0







Interpret s as slack variables, get dual of LP:

min
x

cTx

s.t. Ax ≥ b

x ≥ 0







[P ] −→
max

y
yb

s.t. yA ≤ cT

y ≥ 0







[D]

Alternative derivation of LP dual

Lagrangian dual⇒ find tightest lower bound for LP

min cTx s.t. Ax ≥ b and x ≥ 0

Multiply constraints Ax ≥ b by coefficients y1, . . . , ym to
obtain the inequalities yiAx ≥ yib, valid provided y ≥ 0

Sum over i:
∑

i yiAx ≥∑i yib = yAx ≥ yb

Look for y such that obtained inequalities are as

stringent as possible whilst still a lower bound for cTx

⇒ yb ≤ yAx and yb ≤ cTx

Suggests setting yA = cT and maximizing yb

Obtain LP dual: max yb s.t. yA = cT and y ≥ 0.

Strong Duality for LP

Thm.

If x is optimum of a linear problem and y is the optimum
of its dual, primal and dual objective functions attain the
same values at x and respectively y.

Proof
Assume x optimum, KKT conditions hold

Recall (KKT2) ∀j ≤ n(sixi = 0),
∀i ≤ m (yi(bi − Aix) = 0)

Get y(b− Ax) = sx⇒ yb = (yA + s)x

By (KKT1) yA + s = cT

Obtain yb = cTx

Strong Duality for convex NLPs I

Theory of KKT conditions derived for generic NLP
min f(x) s.t. g(x) ≤ 0, independent of linearity of f, g

Derive strong duality results for convex NLPs

Slater condition ∃x′ ∈ F (P ) (g(x′) < 0) requires
non-empty interior of F (P )

Let f∗ = minx:g(x)≤0 f(x) be the optimal objective

function value of the primal problem P

Let p∗ = maxy≥0 minx∈F (P ) L(x, y) be the optimal

objective function value of the Lagrangian dual

Thm.

If f, g are convex functions and Slater’s condition holds,
then f∗ = p∗.

Strong Duality for convex NLPs II
Proof

- Let A = {(λ, t) | ∃x (λ ≥ g(x) ∧ t ≥ f(x))}, B = {(0, t) | t < f∗}

- A =set of values taken by

constraints and objectives

- A ∩ B = ∅ for otherwise f∗ not

optimal

- P is convex⇒ A,B convex

- ⇒ ∃ separating hyperplane

uλ + µt = α s.t.

∀(λ, t) ∈ A (uλ + µt ≥ α) (4)

∀(λ, t) ∈ B (uλ + µt ≤ α) (5)

- Since λ, t may increase indefinitely, (4) bounded below⇒ u ≥ 0, µ ≥ 0

Strong Duality for convex NLPs III
Proof

- Since λ = 0 in B, (5)⇒ ∀t < f∗ (µt ≤ α)

- Combining latter with (4) yields

∀x (ug(x) + µf(x) ≥ µf∗) (6)

- Suppose µ = 0: (6) becomes ug(x) ≥ 0 for all feasible x; by Slater’s

condition ∃x′ ∈ F (P ) (g(x′) < 0), so u ≤ 0, which together with u ≥ 0

implies u = 0; hence (u, µ) = 0 contradicting separating hyperplane

theorem, thus µ > 0

- Setting µy = u in (6) yields ∀x ∈ F (P ) (f(x) + yg(x) ≥ f∗)

- Thus, for all feasible x we have L(x, y) ≥ f∗

- In particular, p∗ = maxy minx L(x, y) ≥ f∗

- Weak duality implies p∗ ≤ f∗

- Hence, p∗ = f∗

Rules for LP dual

Primal Dual

min max

variables x constraints

constraints variables y

objective coefficients c constraint right hand sides c

constraint right hand sides b objective coefficients b

Aix ≥ bi yi ≥ 0

Aix ≤ bi yi ≤ 0

Aix = bi yi unconstrained

xj ≥ 0 yAj ≤ cj

xj ≤ 0 yAj ≥ cj

xj unconstrained yAj = cj

Ai: i-th row of A Aj: j-th column of A

Examples: LP dual formulations

Primal problem P and canonical form:

max
x1,x2

x1 + x2

s.t. x1 + 2x2 ≤ 2

2x1 + x2 ≤ 2

x ≥ 0







⇒
−min

x1,x2

−x1 − x2

s.t. −x1 − 2x2 ≥ −2

−2x1 − x2 ≥ −2

x ≥ 0







Dual problem D and reformulation:

−max
y1,y2

−2y1 − 2y2

s.t. −y1 − 2y2 ≤ −1

−2y1 − y2 ≤ −1

y ≥ 0







⇒
min
y1,y2

2y1 + 2y2

s.t. y1 + 2y2 ≥ 1

2y1 + y2 ≥ 1

y ≥ 0

Example: Shortest Path Problem

SHORTEST PATH PROBLEM.
Input: weighted digraph G =
(V,A, c), and s, t ∈ V .
Output: a minimum-weight path
in G from s to t. 1

2

2

2

6

4
1

1

1

1

5

0

3

s

t

min
x≥0

∑

(u,v)∈A

cuvxuv

∀ v ∈ V
∑

(v,u)∈A

xvu −
∑

(u,v)∈A

xuv =







1 v = s

−1 v = t

0 othw.







[P ]

max
y

ys − yt

∀ (u, v) ∈ A yv − yu ≤ cuv

}

[D]

Shortest Path Dual

2

2

4

1

2

3

4

cols (1,2) (1,3) . . . (4,1) . . .

rows\c 2 2 . . . 4 . . . b

1 1 1 . . . -1 . . . 0 y1

2 -1 0 . . . 0 . . . 0 y2

3 0 -1 . . . 0 . . . 0 y3

4 0 0 . . . 1 . . . 0 y4
...

...
...

...
...

...

s 0 0 . . . 0 . . . 1 ys

...
...

...
...

...
...

t 0 0 . . . 0 . . . -1 yt

...
...

...
...

...
...

x12 x13 . . . x41 . . .

SP Mechanical Algorithm

1
2

3

ysyt
min yt max ys

st

≤ c13

1 2

3

ysyt

st
= c1t

= c21 = cs2

xuv = 1

KKT2 on [D] ⇒ ∀(u, v) ∈ A (xuv(yv − yu − cuv) = 0)⇒
∀(u, v) ∈ A (xuv = 1→ yv − yu = cuv)

exAMPLes

LP example: .mod

# lp.mod

param n integer, default 3;

param m integer, default 4;

set N := 1..n;

set M := 1..m;

param a{M,N};

param b{M};

param c{N};

var x{N} >= 0;

minimize objective: sum{j in N} c[j]*x[j];

subject to constraints{i in M} :

sum{j in N} a[i,j]*x[j] <= b[i];

LP example: .dat

# lp.dat

param n := 3; param m := 4;

param c :=

1 1

2 -3

3 -2.2 ;

param b :=

1 -1

2 1.1

3 2.4

4 0.8 ;

param a : 1 2 3 :=

1 0.1 0 -3.1

2 2.7 -5.2 1.3

3 1 0 -1

4 1 1 0 ;

LP example: .run

# lp.run

model lp.mod;

data lp.dat;

option solver cplex;

solve;

display x;

LP example: output

CPLEX 11.0.1: optimal solution; objective -11.30153846

0 dual simplex iterations (0 in phase I)

x [*] :=

1 0

2 0.8

3 4.04615

;

MILP example: .mod
# milp.mod

param n integer, default 3;

param m integer, default 4;

set N := 1..n;

set M := 1..m;

param a{M,N};

param b{M};

param c{N};

var x{N} >= 0, <= 3, integer;

var y >= 0;

minimize objective: sum{j in N} c[j]*x[j];

subject to constraints{i in M} :

sum{j in N} a[i,j]*x[j] - y <= b[i];

MILP example: .run

# milp.run

model milp.mod;

data lp.dat;

option solver cplex;

solve;

display x;

display y;

MILP example: output

CPLEX 11.0.1: optimal integer solution; objective -15.6

0 MIP simplex iterations

0 branch-and-bound nodes

x [*] :=

1 0

2 3

3 3

;

y = 2.2

NLP example: .mod

# nlp.mod

param n integer, default 3;

param m integer, default 4;

set N := 1..n;

set M := 1..m;

param a{M,N};

param b{M};

param c{N};

var x{N} >= 0.1, <= 4;

minimize objective:

c[1]*x[1]*x[2] + c[2]*x[3]ˆ2 + c[3]*x[1]*x[2]/x[3];

subject to constraints{i in M diff {4}} :

sum{j in N} a[i,j]*x[j] <= b[i]/x[i];

subject to constraint4 : prod{j in N} x[j] <= b[4];

NLP example: .run
# nlp.run

model nlp.mod;

data lp.dat;

## only enable one of the following methods

## 1: local solution

option solver minos;

# starting point

let x[1] := 0.1;

let x[2] := 0.1; # try 0.1, 0.4

let x[3] := 0.2;

## 2: global solution (heuristic)

#option solver bonmin;

## 3: global solution (guaranteed)

#option solver couenne;

solve;

display x;

NLP example: output

MINOS 5.51: optimal solution found.

140 iterations, objective -47.9955

Nonlin evals: obj = 358, grad = 357, constrs = 358,

x [*] :=

1 0.1

2 0.1

3 4

;

With x2 = 0.4 we get 47 iterations, objective −38.03000929
and x = (2.84106, 1.37232, 0.205189).

INF572 2010/11 – p. 86



MINLP example: .mod
# minlp.mod

param n integer, default 3;

param m integer, default 4;

set N := 1..n;

set M := 1..m;

param a{M,N};

param b{M};

param c{N};

param epsilon := 0.1;

var x{N} >= 0, <= 4, integer;

minimize objective:

c[1]*x[1]*x[2] + c[2]*x[3]ˆ2 + c[3]*x[1]*x[2]/x[3] +

x[1]*x[3]ˆ3;

subject to constraints{i in M diff {4}} :

sum{j in N} a[i,j]*x[j] <= b[i]/(x[i] + epsilon);

subject to constraint4 : prod{j in N} x[j] <= b[4];

MINLP example: .run
# minlp.run

model minlp.mod;

data lp.dat;

## only enable one of the following methods:

## 1: global solution (heuristic)

#option solver bonmin;

## 2: global solution (guaranteed)

option solver couenne;

solve;

display x;

MINLP example: output

bonmin: Optimal

x [*] :=

1 0

2 4

3 4

;

Sudoku

INF572 2010/11 – p. 90



Sudoku: problem class

What is the problem class?

The class of all sudoku grids

Replace {1, . . . , 9} with a set K

Will need a set H = {1, 2, 3} to define 3× 3 sub-grids

An “instance” is a partial assignment of integers to
cases in the sudoku grid

We model an empty sudoku grid, and then fix certain
variables at the appropriate values

Modelling the Sudoku

Q: What are the decisions to be taken?

A: Whether to place an integer in K = {1, . . . , 9} in the
case at coordinates (i, j) on the square grid (i, j ∈ K)

We might try integer variables yij ∈ K

Q: What is the objective function?

A: There is no “natural” objective; we might wish to
employ one if needed

Q: What are the constraints?

A: For example, the first row should contain all numbers
in K; hence, we should express a constraint such as:

if y11 = 1 then y1ℓ 6= 1 for all ℓ ≥ 1;

if y11 = 2 then y1ℓ 6= 2 for all ℓ ≥ 2;

. . . (for all values, column and row indices)

Sudoku constraints 1
In other words,

∀i, j, k ∈ K, ℓ 6= j (yij = k → yiℓ 6= k)

Put it another way: a constraint that says “all values should
be different”

In constraint programming (a discipline related to MP) there
is a constraint

∀i ∈ K AllDiff(yij | j ∈ K)

that asserts that all variables in its argument take
different values: we can attempt to implement it in MP

A set of distinct values has the pairwise distinctness property:
∀i, p, q ∈ K yip 6= yiq, which can also be written as:

∀i, p < q ∈ K |yip − yiq| ≥ 1

Sudoku constraints 2

We also need the same constraints in each column:

∀j, p < q ∈ K |ypj − yqj | ≥ 1

. . . and in some appropriate 3× 3 sub-grids:

1. let H = {1, . . . , 3} and α = |K|/|H|; for all h ∈ H
define Rh = {i ∈ K | i > (h− 1)α ∧ i ≤ hα} and
Ch = {j ∈ K | j > (h− 1)α ∧ j ≤ hα}

2. show that for all (h, l) ∈ H ×H, the set Rh × Cl

contains the case coordinates of the (h, l)-th 3× 3
sudoku sub-grid

Thus, the following constraints must hold:

∀h, l ∈ H, i < p ∈ Rh, j < q ∈ Cl |yij − ypq| ≥ 1

INF572 2010/11 – p. 94



The Sudoku MINLP

The whole model is as follows:

min 0

∀i, p < q ∈ K |yip − yiq| ≥ 1

∀j, p < q ∈ K |ypj − yqj | ≥ 1

∀h, l ∈ H, i < p ∈ Rh, j < q ∈ Cl |yij − ypq| ≥ 1

∀i ∈ K, j ∈ K yij ≥ 1

∀i ∈ K, j ∈ K yij ≤ 9

∀i ∈ K, j ∈ K yij ∈ Z







This is a nondifferentiable MINLP

MINLP solvers (BONMIN, MINLP_BB, COUENNE) can’t
solve it

Absolute value reformulation
This MINLP, however, can be linearized:

|a− b| >= 1 ⇐⇒ a− b >= 1 ∨ b− a >= 1

For each i, j, p, q ∈ K we introduce a binary variable
wpq

ij = 1 if yij − ypq ≥ 1 and 0 if ypq − yij ≥ 1

For all i, j, p, q ∈ K we add constraints

1. yij − ypq ≥ 1−M(1− wpq
ij )

2. ypq − yij ≥ 1−Mwpq
ij

where M = |K|+ 1

This means: if wpq
ij = 1 then constraint 1 is active and 2

is always inactive (as ypq − yij is always greater than

−|K|); if wpq
ij = 0 then 2 is active and 1 inactive

Transforms problematic absolute value terms into
added binary variables and linear constraints

The reformulated model
The reformulated model is a MILP:

min 0

∀i, p < q ∈ K yip − yiq ≥ 1−M(1− wiq
ip)

∀i, p < q ∈ K yiq − yip ≥ 1−Mwiq
ip

∀j, p < q ∈ K ypj − yqj ≥ 1−M(1− wqj
pj)

∀j, p < q ∈ K yqj − ypj ≥ 1−Mwqj
pj

∀h, l ∈ H, i < p ∈ Rh, j < q ∈ Cl yij − ypq ≥ 1−M(1− wpq
ij )

∀h, l ∈ H, i < p ∈ Rh, j < q ∈ Cl ypq − yij ≥ 1−Mwpq
ij

∀i ∈ K, j ∈ K yij ≥ 1

∀i ∈ K, j ∈ K yij ≤ 9

∀i ∈ K, j ∈ K yij ∈ Z







It can be solved by CPLEX; however, it has O(|K|4) binary variables

on a |K|2 cases grid with |K| values per case (O(|K|3) in total) —

often an effect of bad modelling

A better model
In such cases, we have to go back to variable definitions

One other possibility is to define binary variables

∀i, j, k ∈ K (xijk = 1) if the (i, j)-th case has value k, and 0

otherwise

Each case must contain exactly one value

∀i, j ∈ K
∑

k∈K

xijk = 1

For each row and value, there is precisely one row that
contains that value, and likewise for cols

∀i, k ∈ K
∑

j∈K

xijk = 1 ∧ ∀j, k ∈ K
∑

i∈K

xijk = 1

Similarly for each Rh × Ch sub-square

∀h, l ∈ H, k ∈ K
∑

i∈Rh,j∈Cl

xijk = 1

The Sudoku MILP
The whole model is as follows:

min 0

∀i ∈ K, j ∈ K
∑

k∈K

xijk = 1

∀i ∈ K, k ∈ K
∑

j∈K

xijk = 1

∀j ∈ K, k ∈ K
∑

i∈K

xijk = 1

∀h ∈ H, l ∈ H, k ∈ K
∑

i∈Rh,j∈Cl

xijk = 1

∀i ∈ K, j ∈ K, k ∈ K xijk ∈ {0, 1}







This is a MILP with O(|K|3) variables

Notice that there is a relation ∀i, j ∈ K yij =
∑

k∈K

kxijk between the

MINLP and the MILP

The MILP variables have been derived by the MINLP ones by “disaggregation”

Sudoku model file

param Kcard integer, >= 1, default 9;

param Hcard integer, >= 1, default 3;

set K := 1..Kcard;

set H := 1..Hcard;

set R{H};

set C{H};

param alpha := card(K) / card(H);

param Instance {K,K} integer, >= 0, default 0;

let {h in H} R[h] := {i in K : i > (h-1) * alpha and i <= h * alpha};

let {h in H} C[h] := {j in K : j > (h-1) * alpha and j <= h * alpha};

var x{K,K,K} binary;

minimize nothing: 0;

subject to assignment {i in K, j in K} : sum{k in K} x[i,j,k] = 1;

subject to rows {i in K, k in K} : sum{j in K} x[i,j,k] = 1;

subject to columns {j in K, k in K} : sum{i in K} x[i,j,k] = 1;

subject to squares {h in H, l in H, k in K} :

sum{i in R[h], j in C[l]} x[i,j,k] = 1;

Sudoku data file

param Instance :=

1 1 2 1 9 1 2 2 4 2 3 1 2 4 9

2 6 2 2 7 8 2 8 6 3 1 5 3 2 8

3 8 2 3 9 7 4 4 5 4 5 1 4 6 3

5 5 9 6 4 7 6 5 8 6 6 6 7 1 3

7 2 2 7 3 6 7 8 4 7 9 9 8 2 1

8 3 9 8 4 4 8 6 5 8 7 2 8 8 8

9 1 8 9 9 6 ;

Sudoku run file

# sudoku

# replace "/dev/null" with "nul" if using Windows

option randseed 0;

option presolve 0;

option solver_msg 0;

model sudoku.mod;

data sudoku-feas.dat;

let {i in K, j in K : Instance[i,j] > 0} x[i,j,Instance[i,j]] := 1;

fix {i in K, j in K : Instance[i,j] > 0} x[i,j,Instance[i,j]];

display Instance;

option solver cplex;

solve > /dev/null;

param Solution {K, K};

if (solve_result = "infeasible") then {

printf "instance is infeasible\n";

} else {

let {i in K, j in K} Solution[i,j] := sum{k in K} k * x[i,j,k];

display Solution;

}

Sudoku AMPL output

liberti@nox$ cat sudoku.run | ampl

Instance [*,*]

: 1 2 3 4 5 6 7 8 9 :=

1 2 0 0 0 0 0 0 0 1

2 0 4 1 9 0 2 8 6 0

3 5 8 0 0 0 0 0 2 7

4 0 0 0 5 1 3 0 0 0

5 0 0 0 0 9 0 0 0 0

6 0 0 0 7 8 6 0 0 0

7 3 2 6 0 0 0 0 4 9

8 0 1 9 4 0 5 2 8 0

9 8 0 0 0 0 0 0 0 6

;

instance is infeasible

Sudoku data file 2

But with a different data file. . .

param Instance :=

1 1 2 1 9 1 2 2 4 2 3 1 2 4 9

2 6 2 2 7 8 2 8 6 3 1 5 3 2 8

3 8 2 3 9 7 4 4 5 4 5 1 4 6 3

5 5 9 6 4 7 6 5 8 6 6 6 7 1 3

7 2 2 7 8 4 7 9 9 8 2 1

8 3 9 8 4 4 8 6 5 8 7 2 8 8 8

9 1 8 9 9 6 ;

Sudoku data file 2 grid

. . . corresponding to the grid below. . .

2 1

4 1 9 2 8 6

5 8 2 7

5 1 3

9

7 8 6

3 2 4 9

1 9 4 5 2 8

8 6

Sudoku AMPL output 2

. . . we find a solution!

liberti@nox$ cat sudoku.run | ampl

Solution [*,*]

: 1 2 3 4 5 6 7 8 9 :=

1 2 9 6 8 5 7 4 3 1

2 7 4 1 9 3 2 8 6 5

3 5 8 3 6 4 1 9 2 7

4 4 7 8 5 1 3 6 9 2

5 1 6 5 2 9 4 3 7 8

6 9 3 2 7 8 6 1 5 4

7 3 2 7 1 6 8 5 4 9

8 6 1 9 4 7 5 2 8 3

9 8 5 4 3 2 9 7 1 6

;

Kissing Number Problem

KNP: problem class

What is the problem class?

There is no number in the problem definition:

How many unit balls with disjoint interior can be placed

adjacent to a central unit ball in Rd?

Hence the KNP is already defined as a problem class

Instances are given by assigning a positive integer to
the only parameter d

INF572 2010/11 – p. 108



Modelling the KNP
Q: What are the decisions to be taken?

A: How many spheres to place, and where to place them

For each sphere, two types of variables

1. a logical one: yi = 1 if sphere i is present, and 0 otherwise

2. a d-vector of continuous ones: xi = (xi1, . . . , xid), position of i-th

sphere center

Q: What is the objective function?

A: Maximize the number of spheres

Q: What are the constraints?

A: Two types of constraints

1. the i-th center must be at distance 2 from the central sphere if the

i-th sphere is placed (center constraints)

2. for all distinct (and placed) spheres i, j, for their interior to be

disjoint their centers must be at distance ≥ 2 (distance constraints)

Assumptions
1. Logical variables y

Since the objective function counts the number of placed

spheres, it must be something like
∑

i yi

What set N does the index i range over?

Denote k∗(d) the optimal solution to the KNP in RD

Since k∗(d) is unknown a priori, we cannot know N a priori;

however, without N , we cannot express the objective function

Assume we know an upper bound k̄ to k∗(d); then we can define

N = {1, . . . , k̄} (and D = {1, . . . , d})
2. Continuous variables x

Since any sphere placement is invariant by translation, we assume

that the central sphere is placed at the origin

Thus, each continuous variable xik (i ∈ N, k ∈ D) cannot attain

values outside [−2, 2] (why?)

Limit continuous variables: −2 ≤ xik ≤ 2

Problem restatement

The above assumptions lead to a problem restatement

Given a positive integer k̄, what is the maximum

number (smaller than k̄) of unit spheres with dis-

joint interior that can be placed adjacent to a unit

sphere centered at the origin of Rd?

Each time assumptions are made for the sake of modelling, one

must always keep track of the corresponding changes to the

problem definition

The Objective function can now be written as:

max
∑

i∈N

yi

Constraints

Center constraints:

∀i ∈ N ||xi|| = 2yi

(if sphere i is placed then yi = 1 and the constraint
requires ||xi|| = 2, otherwise ||xi|| = 0, which implies
xi = (0, . . . , 0))

Distance constraints:

∀i ∈ N, j ∈ N : i 6= j ||xi − xj || ≥ 2yiyj

(if spheres i, j are both are placed then yiyj = 1 and the

constraint requires ||xi − xj || ≥ 2, otherwise

||xi − xj || ≥ 0 which is always by the definition of norm)

KNP model

max
∑

i∈N

yi

∀i ∈ N
√∑

k∈D

x2
ik = 2yi

∀i ∈ N, j ∈ N : i 6= j
√∑

k∈D

(xik − xjk)2 ≥ 2yiyj

∀i ∈ N yi ≥ 0

∀i ∈ N yi ≤ 1

∀i ∈ N, k ∈ D xik ≥ −2

∀i ∈ N, k ∈ D xik ≤ 2

∀i ∈ N yi ∈ Z







For brevity, we shall write yi ∈ {0, 1} and xik ∈ [−2, 2]

Reformulation 1

Solution times for NLP/MINLP solvers often also
depends on the number of nonlinear terms

We square both sides of the nonlinear constraints, and

notice that since yi are binary variables, y2
i = yi for all

i ∈ N ; we get:

∀i ∈ N
∑

k∈D

x2
ik = 4yi

∀i 6= j ∈ N
∑

k∈D

(xik − xjk)
2 ≥ 4yiyj

which has fewer nonlinear terms than the original
problem

Reformulation 2

Distance constraints are called reverse convex (because if
we replace ≥ with ≤ the constraints become convex);
these constraints often cause solution times to lengthen
considerably

Notice that distance constraints are repeated when i, j
are swapped

Change the quantifier to i ∈ N, j ∈ N : i < j reduces the
number of reverse convex constraints in the problem;
get:

∀i ∈ N
∑

k∈D

x2
ik = 4yi

∀i < j ∈ N
∑

k∈D

(xik − xjk)
2 ≥ 4yiyj

KNP model revisited

max
∑

i∈N

yi

∀i ∈ N
∑

k∈D

x2
ik = 4yi

∀i ∈ N, j ∈ N : i < j
∑

k∈D

(xik − xjk)
2 ≥ 4yiyj

∀i ∈ N, k ∈ D xik ∈ [−2, 2]

∀i ∈ N yi ∈ {0, 1}







This formulation is a (nonconvex) MINLP

KNP model file

# knp.mod

param d default 2;

param kbar default 7;

set D := 1..d;

set N := 1..kbar;

var y{i in N} binary;

var x{i in N, k in D} >= -2, <= 2;

maximize kstar : sum{i in N} y[i];

subject to center{i in N} : sum{k in D} x[i,k]ˆ2 = 4*y[i];

subject to distance{i in N, j in N : i < j} :

sum{k in D} (x[i,k] - x[j,k])ˆ2 >= 4*y[i]*y[j];

KNP data file

Since the only data are the parameters d and k̄ (two
scalars), for simplicity we do not use a data file at all, and
assign values in the model file instead

KNP run file

# knp.run

model knp.mod;

option solver couenne;

let kbar := 12;

let d := 3;

solve;

display x,y;

display kstar;

KNP solution (?)

We tackle the easiest possible KNP instance (d = 2),

and give it an upper bound k̄ = 7

It is easy to see that k∗(2) = 6 (place 6 circles adjacent
to another circle in an exagonal lattice)

Yet, after several minutes of CPU time COUENNE has not
made any progress from the trivial feasible solution
y = 0, x = 0

Likewise, heuristic solvers such as BONMIN and
MINLP BB only find the trivial zero solution and exit

What do we do next?

In order to solve the KNP and deal with other difficult
MINLPs, we need more advanced techniques

Some useful MP theory

Open sets

In general, MP cannot directly model problems involving
sets which are not closed in the usual topology (such as
e.g. open intervals)

The reason is that the minimum/maximum of a
non-closed set might not exist

E.g. what is min
x∈(0,1)

x? Since (0, 1) has no minimum (for

each δ ∈ (0, 1), δ
2 < δ and is in (0, 1)), the question has

no answer

This is why the MP language does not allow writing constraints that

involve the <, > and 6= relations

Sometimes, problems involving open sets can be
reformulated exactly to problems involving closed sets

(e.g. x > 0⇔ x ≥ e−y)

Best fit hyperplane 1

Consider the following problem:

Given m points p1, . . . , pm ∈ Rn, find the hyperplane

w1x1 + · · ·+ wnxn = w0 minimizing the piecewise linear form

f(p, w) =
∑

i∈P

| ∑
j∈N

wjpij − w0|

Mathematical programming formulation:

1. Sets: P = {1, . . . , m}, N = {1, . . . , n}
2. Parameters: ∀i ∈ P pi ∈ Rn

3. Decision variables: ∀j ∈ N wj ∈ R, w0 ∈ R

4. Objective: minw f(p, w)

5. Constraints: none

Trouble: w = 0 is the obvious, trivial solution of no interest

We need to enforce a constraint (w1, . . . , wn, w0) 6= (0, . . . , 0)

Bad news: Rn+1 r {(0, . . . , 0)} is not a closed set

Best fit hyperplane 2

We can implicitly impose such a constraint by transforming the

objective function to minw
f(p,w)
||w|| (for some norm || · ||)

This implies that w is nonzero but the feasible region is Rn+1, which

is both open and closed

Obtain fractional objective — difficult to solve

Suppose w
∗ = (w∗, w∗

0) ∈ Rn+1 is an optimal solution to the above

problem

Then for all d > 0, f(dw∗, p) = df(w∗, p)

Hence, it suffices to determine the optimal direction of w∗, because the

actual vector length simply scales the objective function value

Can impose constraint ||w|| = 1 and recover original objective

Solve reformulated problem:

min{f(w, p) | ||w|| = 1}

Best fit hyperplane 3

The constraint ||w|| = 1 is a constraint schema: we must
specify the norm

Some norms can be reformulated to linear constraints,
some cannot

max-norm (l∞) 2-sphere (square), Euclidean norm (l2)
2-sphere (circle), abs-norm (l1) 2-sphere (rhombus)

max- and abs-norms are piecewise linear, they can be
linearized exactly by using binary variables (see later)

Convexity in practice

Recognizing whether an arbitrary function is convex is
an undecidable problem

For some functions, however, this is possible

Certain functions are known to be convex (such as all

affine functions, cx2n for n ∈ N and c ≥ 0, exp(x),
− log(x))

Norms are convex functions

The sum of two convex functions is convex

Application of the above rules repeatedly sometimes
works (for more information, see Disciplined Convex
Programming [Grant et al. 2006])

Warning: problems involving integer variables are in general not

convex; however, if the objective function and constraints are convex

forms, we talk of convex MINLPs

Recognizing convexity 1

Consider the following mathematical program

min
x,y∈[0,10]

8x2 − 17xy + 10y2

x− y ≥ 1

x2y ≥ 1

Objective function and constraints contain nonconvex
term xy

There is no reason to believe that x2y ≥ 1 might be
convex

Is this problem convex or not?

Recognizing convexity 2

The objective function can be written as (x, y)TQ(x, y)

where Q =

(

8 −8

−9 10

)

The eigenvalues of Q are 9±
√

73 (both positive), hence
the Hessian of the objective is positive definite, hence
the objective function is convex

The affine constraint x− y ≥ 1 is convex by definition

x2y ≥ 1 is not, but can be reformulated:

1. Take logarithms of both sides: log x2y ≥ log 1

2. Implies 2 log x + log y ≥ 0⇒ −2 log x− log y ≤ 0

3. − log is a convex function, sum of convex functions is
convex, convex ≤ affine is a convex constraint

Recognizing convexity 3

Indeed, the set {(x, y) | x2y ≥ 1} is shown in yellow below
the surface

Both pictures represent the same set

Recognizing convexity 4

model;

var x <= 10, >= 0.1;

var y <= 10, >= 0.1;

minimize f: 8*xˆ2 -17*x*y + 10*yˆ2;

subject to c1: x-y >= 1;

subject to c2: xˆ2*y >= 1;

option solver_msg 0;

printf "solving with sBB (couenne)\n";

option solver couenne;

solve > /dev/null;

display x,y;

printf "solving with local NLP solver (ipopt)\n";

option solver ipopt; let x := 0.1; let y := 0.1;

solve > /dev/null; display x,y;

Get approx. same solution (1.5, 0.5) from COUENNE and IPOPT

Total Unimodularity

A matrix A is Totally Unimodular (TUM) if all invertible
square submatrices of A have determinant ±1
Thm.

If A is TUM, then all vertices of the polyhedron

{x ≥ 0 | Ax ≤ b}

have integral components

Consequence: if the constraint matrix of a given MILP is
TUM, then it suffices to solve the relaxed LP to get a
solution for the original MILP

An LP solver suffices to solve the MILP to optimality

TUM in practice 1

If A is TUM, AT and (A|I) are TUM

TUM Sufficient conditions. An m× n matrix A is TUM if:

1. for all i ≤ m, j ≤ n we have aij ∈ {0, 1,−1};
2. each column of A contains at most 2 nonzero

coefficients;

3. there is a partition R1, R2 of the set of rows such that
for each column j,

∑

i∈R1
aij −

∑

i∈R2
aij = 0.

Example: take R1 = {1, 3, 4}, R2 = {2}







1 0 1 1 0 0

0 −1 0 1 −1 1

−1 −1 0 0 0 1

0 0 −1 0 −1 0

TUM in practice 2

Consider digraph G = (V,A) with nonnegative variables
xij ∈ R+ defined on each arc

Flow constraints ∀i ∈ V
∑

(i,j)∈A

xij −
∑

(j,i)∈A

xji = bi yield a

TUM matrix (partition: R1 = all rows, R2 = ∅— prove it)

Maximum flow problems can be solved to integrality by
simply solving the continuous relaxation with an LP
solver

The constraints of the set covering problem do not form a TUM. To

prove this, you just need to find a counterexample

Maximum flow problem

Given a network on a directed graph G = (V,A) with a
source node s, a destination node t, and integer capacities
uij on each arc (i, j). We have to determine the maximum

integral amount of material flow that can circulate on the
network from s to t. The variables xij ∈ Z, defined for each

arc (i, j) in the graph, denote the number of flow units.

maxx

∑

(s,i)∈A

xsi

∀ i ≤ V,
i 6= s

i 6= t

∑

(i,j)∈A

xij =
∑

(j,i)∈A

xji

∀(i, j) ∈ A 0 ≤ xij ≤ uij

∀(i, j) ∈ A xij ∈ Z







3
1

4

i

xji

xij

Max Flow Example 1

1 2 3

4 5 6
7

4

5

2
1

7

6

5

1

1

3

2

7

arc capacities as shown in italics: find the maximum flow
between node s = 1 and t = 7

Max Flow: MILP formulation

Sets: V = {1, . . . , n}, A ⊆ V × V

Parameters: s, t ∈ V , u : A→ R+

Variables: x : A→ Z+

Objective: max
∑

(s,i)∈A

xsi

Constraints: ∀i ∈ V r {s, t} ∑

(i,j)∈A

xij =
∑

(j,i)∈A

xji

Max Flow: .mod file

# maxflow.mod

param n integer, > 0, default 7;

param s integer, > 0, default 1;

param t integer, > 0, default n;

set V := 1..n;

set A within {V,V};

param u{A} >= 0;

var x{(i,j) in A} >= 0, <= u[i,j], integer;

maximize flow : sum{(s,i) in A} x[s,i];

subject to flowcons{i in V diff {s,t}} :

sum{(i,j) in A} x[i,j] = sum{(j,i) in A} x[j,i];

Max Flow: .dat file

# maxflow.dat

param : A : u :=

1 2 5

1 4 4

1 5 1

2 4 2

3 2 1

3 4 1

3 7 2

4 5 7

5 3 6

5 6 5

6 2 3

6 7 7 ;

Max Flow: .run file

# maxflow.run

model maxflow.mod;

#model maxflow_constrained.mod;

data maxflow.dat;

option solver_msg 0;

option solver cplex;

solve;

for {(i,j) in A : x[i,j] > 0} {

printf "x[%d,%d] = %g\n", i,j,x[i,j];

}

display flow;

Max Flow: MILP solution

5 units of flow

6 units of flow

1 2 3

4 5 6
7

4

5

2
1

7

6

5

1

1

3

2

7

1unit of flow

2 units of flow

4 units of flow maximum flow = 7

x[1,2] = 2

x[1,4] = 4

x[1,5] = 1

x[2,4] = 2

x[3,7] = 2

x[4,5] = 6

x[5,3] = 2

x[5,6] = 5

x[6,7] = 5

flow = 7

Max Flow: LP solution

Relax integrality constraints (take away integer keyword)

5 units of flow

6 units of flow

1 2 3

4 5 6
7

4

5

2
1

7

6

5

1

1

3

2

7

1unit of flow

2 units of flow

4 units of flow maximum flow = 7

Get the same solution

Reformulations

Reformulations

If problems P,Q are related by a computable function f
through the relation f(P,Q) = 0, Q is an auxiliary problem

with respect to P .

Exact reformulations: preserve all optimality properties

Narrowings: preserve some optimality properties

Relaxations: provide bounds to the optimal objective
function value

Approximations: formulation Q depending on a
parameter k such that “ lim

k→∞
Q(k)” is an exact

reformulation, narrowing or relaxation

INF572 2010/11 – p. 144



Exact reformulations

P

Q

F

F LL

GG

φ

φ|L

φ|G

Main idea: if we find an optimum of Q, we can map it back to the same

type of optimum of P , and for all optima of P , there is a correspond-

ing optimum in Q. Also known as exact reformulation

Narrowings

P

Q

F

F

G

G

φ

φ|G

Main idea: if we find a global optimum of Q, we can map
it back to a global optimum of P . There may be optima
of P without a corresponding optimum in Q.

Relaxations

A problem Q is a relaxation of P if the globally optimal
value of the objective function min fQ of Q is a lower

bound to that of P .

Approximations

Q is an approximation of P if there exist: (a) an auxiliary problem

Q∗ of P ; (b) a sequence {Qk} of problems; (c) an integer ℓ > 0;

such that:

1. Q = Qℓ

2. ∀ objective f∗ in Q∗ there is a sequence of objectives fk of Qk

converging uniformly to f∗;

3. ∀ constraint l∗i ≤ g∗i (x) ≤ u∗
i of Q∗ there is a sequence of constraints

lki ≤ gk
i (x) ≤ uk

i of Qk such that gk
i converges uniformly to g∗i , lki

converges to l∗i and uk
i to u∗

i

There can be approximations to exact reformulations, narrowings,

relaxations.

 Q1, Q2, Q3, Qℓ, Q∗ P. . .. . .

approximation of P

〈auxiliary problem of〉

Fundamental results

Exact reformulation, narrowing, relaxation,
approximation are all transitive relations

An approximation of any type of reformulation is an approximation

A reformulation consisting of exact reformulations,
narrowings, relaxations is a relaxation

A reformulation consisting of exact reformulations and narrowings is

a narrowing

A reformulation consisting of exact reformulations is an
exact reformulation

exact reform.exact reform. narrowingsrelaxations approximations

Reformulations in practice

Reformulations are used to transform problems into
equivalent (or related) formulations which are somehow
“better”

Basic reformulation operations :

1. change parameter values

2. add / remove variables

3. adjoin / remove constraints

4. replace a term with another term (e.g. a product xy
with a new variable w)

Product of binary variables

Consider binary variables x, y and a cost c to be added
to the objective function only of xy = 1

⇒ Add term cxy to objective

Problem becomes mixed-integer (some variables are
binary) and nonlinear

Reformulate “xy” to MILP form (PRODBIN reform.):

replace xy by z

add z ≤ y , z ≤ x

z ≥ 0, z ≥ x + y − 1

x, y ∈ {0, 1} ⇒
z = xy

Application to the KNP
In the RHS of the KNP’s distance constraints we have 4yiyj , where

yi, yj are binary variables

We apply PRODBIN (call the added variable wij):

min
P

i∈N

yi

∀i ∈ N
P

k∈D

x2
ik

= 4yi

∀i ∈ N, j ∈ N : i < j
P

k∈D

(xik − xjk)2 ≥ 4wij

∀i ∈ N, j ∈ N : i < j wij ≤ yi

∀i ∈ N, j ∈ N : i < j wij ≤ yj

∀i ∈ N, j ∈ N : i < j wij ≥ yi + yj − 1

∀i ∈ N, j ∈ N : i < j wij ∈ [0, 1]

∀i ∈ N, k ∈ D xik ∈ [−2, 2]

∀i ∈ N yi ∈ {0, 1}

9

>

>

>

>

>

>

>

>

>

>

>

>

>

>

>

>

>

>

>

>

>

>

=

>

>

>

>

>

>

>

>

>

>

>

>

>

>

>

>

>

>

>

>

>

>

;

Still a MINLP, but fewer nonlinear terms

Still numerically difficult (2h CPU time to find k∗(2) ≥ 5)

Product of bin. and cont. vars.
PRODBINCONT reformulation

Consider a binary variable x and a continuous variable

y ∈ [yL, yU ], and assume product xy is in the problem

Replace xy by an added variable w

Add constraints:

w ≤ yUx

w ≥ yLx

w ≤ y + yL(1− x)

w ≥ y − yU (1− x)

Exercise 1 : show that PRODBINCONT is an exact reformulation

Exercise 2 : show that if y ∈ {0, 1} then PRODBINCONT is equivalent to

PRODBIN

Prod. cont. vars.: approximation

BILINAPPROX approximation

Consider x ∈ [xL, xU ], y ∈ [yL, yU ] and product xy

Suppose xU − xL ≤ yU − yL, consider an integer d > 0

Replace [xL, xU ] by a finite set

D = {xL + (i− 1)γ | 1 ≤ i ≤ d}, where γ = xU−xL

d−1

→

BILINAPPROX

Replace the product xy by a variable w

Add binary variables zi for i ≤ d

Add assignment constraint for zi’s

∑

i≤d

zi = 1

Add definition constraint for x:

x =
∑

i≤d

(xL + (i− 1)γ)zi

(x takes exactly one value in D)

Add definition constraint for w

w =
∑

i≤d

(xL + (i− 1)γ)ziy (7)

Reformulate the products ziy via PRODBINCONT

BILINAPPROX2

BILINAPPROX2 : problem P has a term xy where x ∈
[xL, xU ], y ∈ [yL, yU ] are continuous; assume xU −
xL ≤ yU − yL

1. choose integer k > 0; add q = {qi | 0 ≤ i ≤ k}
to P so that q0 = xL, qk = xU , qi < qi+1 for all i

2. add continuous variable w ∈ [wL, wU ] (com-

puted from ranges of x, y by interval arithmetic)

and replace term xy by w

3. add binary variables zi for 1 ≤ i ≤ k and con-

straint
∑

i≤k zi = 1

4. for all 1 ≤ i ≤ k add constraints:

k → ∞: get iden-

tity (exact) reformu-

lation

k∑

j=1

qj−1zj ≤ xi ≤
k∑

j=1

qjzj

qi+qi−1

2 y − (wU − wL)(1− zi) ≤ w ≤ qi+qi−1

2 y + (wU − wL)(1− zi),

Relaxing bilinear terms

RRLTRELAX : quadratic problem P with terms xixj (i < j) and constrs

Ax = b (x can be bin, int, cont); perform exact reformulation RRLT

first:

1. add continuous variables wij (let wi = (wi1, . . . , w1n))

2. replace product xixj with wij (for all i, j)

3. add the reduced RLT (RRLT) system ∀k Awk − bxk = 0

4. find a partition (B, N) of basic/nonbasic variables of ∀k Awk = 0

such that B corresponds to variables with smallest range

5. for all (i, j) ∈ N add constraints wij = xixj (†)
then replace nonlinear constraints (†) with McCormick’s envelopes

wij ≥ max{xL
i xj + xL

j xi − xL
i xL

j , xU
i xj + xU

j xi − xU
i xU

j }

wij ≤ min{xU
i xj + xL

j xi − xU
i xL

j , xL
i xj + xU

j xi − xL
i xU

j }
A

B

C

D

−1
−0.5

 0
 0.5

 1 −1

−0.5

 0

 0.5

 1

−1

−0.5

 0

 0.5

 1

The effect of RRLT is that of using information in Ax = b to eliminate

Linearizing the l∞ norm

INFNORM [Coniglio et al., MSc Thesis, 2007]. P has

vars x ∈ [−1, 1]d and constr. ||x||∞ = 1,
s.t. x∗ ∈ F(P )↔ −x∗ ∈ F(P ) and f(x∗) = f(−x∗).

1. ∀k ≤ d add binary var uk

2. delete constraint ||x||∞ = 1

3. add constraints:

∀k ≤ d xk ≥ 2uk − 1
∑

k≤d

uk = 1.

Narrowing INFNORM(P ) cuts away all optima having
maxk |xk| = 1 with xk < 1 for all k ≤ d

Approximating squares

INNERAPPROXSQ : P has a continuous variable

x ∈ [xL, xU ] and a term x2 appearing as a

convex term in an objective or constraint

1. add parameters n ∈ N, ε = xU−xL

n−1 ,

x̄i = xL + (i− 1)ε for i ≤ n

2. add a continuous variable w ∈ [wL, wU ],

where wL = 0 if xLxU ≤ 0 or

min((xL)2, (xU )2) otherwise and

wU = max((xL)2, (xU )2)

3. replace all occurrences of term x2 with w

4. add constraints

∀i ≤ n w ≥ (x̄i + x̄i−1)x− x̄ix̄i−1.

Replace convex term by piecewise linear ap-

proximation

x

s

n → ∞: get

identity (exact)

reformulation

Conditional constraints
Suppose ∃ a binary variable y and a constraint g(x) ≤ 0
in the problem

We want g(x) ≤ 0 to be active iff y = 1

Compute maximum value that g(x) can take over all x,
call this M

Write the constraint as:

g(x) ≤M(1− y)

This sometimes called the “big M ” modelling technique

Example:

Can replace constraint (7) in BILINAPPROX as follows:

∀i ≤ d −M(1− zi) ≤ w − (xL + (i− 1)γ)y ≤M(1− zi)

where M s.t. w − (xL + (i− 1)γ)y ∈ [−M, M ] for all w, x, y

Symmetry

Example

Consider the problem

min x1 + x2

3x1 + 2x2 ≥ 1

2x1 + 3x2 ≥ 1

x1, x2 ∈ {0, 1}







AMPL code:
set J := 1..2;

var x{J} binary;

minimize f: sum{j in J} x[j];

subject to c1: 3*x[1] + 2*x[2] >= 1;

subject to c2: 2*x[1] + 3*x[2] >= 1;

option solver cplex;

solve;

display x;

The solution (given by

CPLEX) is x1 = 1, x2 = 0

If you swap x1 with x2, you

obtain the same problem, with

swapped constraints

Hence, x1 = 0, x2 = 1 is

also an optimal solution!

Permutations

We can represent permutations by maps N→ N

The permutation of our example is

0

@

1 2
↓ ↓
2 1

1

A

Permutations are usually written as cycles: e.g. for a

permutation

0

@

1 2 3
↓ ↓ ↓
3 1 2

1

A, which sends 1→ 3, 3→ 2 and

2→ 1, we write (1, 3, 2) to mean 1→ 3→ 2(→ 1)

The permutation of our example is (1, 2) — a cycle of
length 2 (also called a transposition, or swap)

Cycles

Cycles can be multiplied together, but the multiplication
is not commutative: (1, 2, 3)(1, 2) = (1, 3) and
(1, 2)(1, 2, 3) = (2, 3)

The identity permutation e fixes all N

Notice (1, 2)(1, 2) = e and (1, 2, 3)(1, 3, 2) = e, so

(1, 2) = (1, 2)−1 and (1, 3, 2) = (1, 2, 3)−1

Cycles are disjoint when they have no common element

Thm. Disjoint cycles commute

Thm. Every permutation can be written uniquely (up to
order) as a product of disjoint cycles

For each permutation π, let Γ(π) be the set of its disjoint
cycles

INF572 2010/11 – p. 164



Groups

A group is a set G together with a multiplication
operation, an inverse operation, and an identity element
e ∈ G, such that:

1. ∀g, h ∈ G (gh ∈ G) (multiplication closure)

2. ∀g ∈ G (g−1 ∈ G) (inverse closure)

3. ∀f, g, h ∈ G ((fg)h = f(gh)) (associativity)

4. ∀g ∈ G (eg = g) (identity )

5. ∀g ∈ G (g−1g = e) (inverse)

The set {e} is a group (denoted by 1) called the trivial

group

The set of all permutations over {1, . . . , n} is a group,
called the symmetric group of order n, and denoted by Sn

For all B ⊆ {1, . . . , n} define Sym(B) as the symmetric
group over the symbols of B

Generators

Given any subset T ⊆ Sn, the smallest group containing
the permutations in T is the group generated by T , denoted
by 〈T 〉
For example, if T = {(1, 2), (1, 2, 3)}, then 〈T 〉 is
{(1), (1, 2), (1, 3), (2, 3), (1, 2, 3), (1, 3, 2)} = S3

For any n ∈ N, 〈(1, . . . , n)〉 is the cyclic group of order n,
denoted by Cn

Cn is commutative, whereas Sn is not

Commutative groups are also called abelian

Thm. 〈(1, 2), (1, . . . , n)〉 = 〈(i, i + 1) | 1 ≤ i < n〉 = Sn

Subgroups and homomorphisms

A subgroup of a group G is a subset H of G which is also a group

(denoted by H ≤ G); e.g. C3 = {e, (1, 2, 3), (1, 3, 2)} is a subgroup of

S3

Given two groups G, H, a map φ : G→ H such that

∀f, g ∈ G ( φ(fg) = φ(f)φ(g) ) is a homomorphism

Kerφ = {g ∈ G | φ(g) = e} is the kernel of φ (Kerφ ≤ G)

Imφ = {h ∈ H | ∃g ∈ G (h = φ(g))} is the image of φ (Imφ ≤ H)

If φ is injective and surjective (i.e. if Kerφ = 1 and Imφ = H), then φ is

an isomorphism, denoted by G ∼= H

Thm.[Lagrange] For all groups G and H ≤ G, |H| divides |G|

Thm.[Cayley] Every finite group is isomorphic to a subgroup of Sn for

some n ∈ N

Normal subgroups

Let H ≤ G; for all g ∈ G, gH = {gh | h ∈ H} and
Hg = {hg | h ∈ H} are in general subsets (not
necessarily subgroups) of G, and in general gH 6= Hg

If ∀g ∈ G (gH = Hg) then H is a normal subgroup of G,
denoted by H ⊳ G (e.g. C3 ⊳ S3)

If H ⊳ G, then {gH | g ∈ G} is denoted by G/H and has
a group structure with multiplication (fH)(gH) = (fg)H,

inverse (gH)−1 = (g−1)H and identity eH = H

For every group homomorphism φ, Kerφ ⊳ G and
G/Kerφ ∼= Imφ

Group actions

Given a group G and a set X, the action of G on X is a
set of mappings αg : X → X for all g ∈ G, such that

αg(x) = (gx) ∈ X for all x ∈ X

Essentially, the action of G on X is the definition of what
happens to x ∈ X when g is applied to it

For example, if X = Rn and G = Sn, a possible action of
G on X is given by gx being the vector x with
components permuted according to g (e.g. if

x = (0.1,−2,
√

2) and g = (1, 2), then gx = (−2, 0.1,
√

2))

Convention: left multiplication if x is a column vector
(αg(x) = gx), right if x is a row vector (αg(x) = xg): treat
g as a matrix

Orbits

If G acts on X ⊆ Rn, for all x ∈ X, Gx = {gx | g ∈ G}
is the orbit of x w.r.t. G

1

2

1 2

2

1

x = (2, 1, 0)

gx=(1,2,0)
g=(1,2)

gx=(0,2,1)
g=(1,2,3)

gx=(0,1,2)
g=(1,3)

gx=(1,0,2)
g=(1,3,2)

gx=(2,0,1)
g=(2,3)

S3(2, 1, 0)
T

Stabilizers

Given Y ⊆ X, the point-wise stabilizer of Y w.r.t. G is a
subgroup H ≤ G such that hy = y for all h ∈ H, y ∈ Y

pointwise

y1y1 y2y2

y3y3

y4y4

h

Y

X

setwise

y1y1 y2y2

y3y3

y4y4

h

Y

X

The set-wise stabilizer of Y w.r.t. G is a subgroup H ≤ G
such that HY = Y (denote H by stab(Y,G))

Let π ∈ Sn with disjoint cycle product σ1 · · ·σk and
N ⊆ {1 . . . , n}
π[N ] =

∏

σ∈Γ(π)∩Sym(N)

σ: restriction of π to N

Groups and graphs

Given a digraph G = (V,A) with V = {v1, . . . , vn}, the
action of π ∈ Sn on G is the natural action of π on V

π is a graph automorphism if ∀(i, j) ∈ A (π(i), π(j)) ∈ A

For example:

G1

1 2

34
G2

3 2

14
G3

4 1

23

G2 = (1, 3)G1 is a graph automorphism of G1

G3 = (1, 2, 3, 4)G1 is not an automorphism of G1: e.g. (4, 2) ∈ A but

(π(4), π(2)) = (1, 3) 6∈ A

The automorphism group of G1 is 〈e, (1, 3)〉 ∼= C2 (denoted
by Aut(G1))

Back to MP: Symmetries and BB

Symmetries are bad for Branch-and-Bound techniques:
many branches will contain (symmetric) optimal
solutions and therefore will not be pruned by bounding
⇒ deep and large BB trees

-Inf

1.5 2

2 2

← BB tree for symmetric

problem

BB tree for “problem

modulo symmetries”→

-Inf

2 2

How do we write a “mathematical programming formu-
lation modulo symmetries”?

Solution symmetries
The set of solutions of the following problem:

min x11 +x12 +x13 +x21 +x22 +x23

x11 +x12 +x13 ≥ 1

x21 +x22 +x23 ≥ 1

x11 +x21 ≥ 1

x12 +x22 ≥ 1

x13 +x23 ≥ 1

is G(P ) =
{(0, 1, 1, 1, 0, 0), (1, 0, 0, 0, 1, 1), (0, 0, 1, 1, 1, 0),

(1, 1, 0, 0, 0, 1), (1, 0, 1, 0, 1, 0), (0, 1, 0, 1, 0, 1)}

G∗ = stab(G(P ), Sn) is the solution group (variable permutations keeping

G(P ) fixed)

For the above problem, G∗ is

〈(2, 3)(5, 6), (1, 2)(4, 5), (1, 4)(2, 5)(3, 6)〉 ∼= D12

For all x∗ ∈ G(P ), G∗x∗ = G(P )⇒ ∃ only 1 orbit⇒ ∃ only one

solution in G(P ) (modulo symmetries)

How do we find G∗ before solving P?

INF572 2010/11 – p. 174



Formulation symmetries
Cost vector c = (1, 1, 1, 1, 1, 1): cS6 = {c}
RHS vector b = (1, 1, 1, 1, 1): S5b = {b}
Constraint matrix A (constraint order independence⇒ can

always permute rows arbitrarily):

1   1   1   0   0   0

0   0   0   1   1   1

1   0   0   1   0   0

0   1   0   0   1   0

0   0   1   0   0   1

(1,2) (4,5)

0   0   0   1   1   1

0   0   1   0   0   1

1   1   1   0   0   0

1   0   0   1   0   0

0   1   0   0   1   0

1   1   1   0   0   0

0   0   0   1   1   1

0   0   1   0   0   1

0   1   0   0   1   0

1   0   0   1   0   0
(3,4)

=

⇒ (3, 4)A(1, 2)(4, 5) = A

For general LPs with data A, b, c, if
∃π ∈ Sn, σ ∈ Sm (cπ = c ∧ σb = b ∧ σ(Aπ) = A) then π
fixes the formulation of the LP

The MILP formulation group

If P is an LP with data A, b, c, then

GP ={π ∈ Sn | ∃σ ∈ Sm(cπ = c ∧ σb = b ∧ σAπ = A)} (8)

is the formulation group of P

For the example, Gexample
∼= D12

∼= G∗

Thm.

If P is an LP, then GP ≤ G∗
P .

Result can be extended to all MILPs [Margot 2002,
2003, 2007]

Symmetries in MINLPs
Consider the following MINLP P :

min f(x)

g(x) ≤ 0

x ∈ X.







(9)

where X may contain integrality constraints on x

For a row permutation σ ∈ Sm and a column
permutation π ∈ Sn, we define σPπ as follows:

min f(xπ)

σg(xπ) ≤ 0

xπ ∈ X.







(10)

Define ḠP = {π ∈ Sn | ∃σ ∈ Sm (σPπ = P )}

A computable definition
Establishing whether ∀x (σAxπ = Ax) is easy, just look at

components of A and σAπ

In general, the statement ∀x (σg(xπ) = g(x) ∧ f(xπ) = f(x)) is

undecidable

Assume we have a computable “equality oracle” equal(h1, h2) so

that:

if equal(h1, h2) =true, then ∀x (h1(x) = h2(x))

The converse may not hold

Define GP as ḠP with = replaced by equal returning true

Can show GP ≤ ḠP ≤ G∗
P

Decision problems:

FORMULATION SYMMETRY. Given formulations P, Q and the oracle equal, are

there permutations σ, π such that P = σQπ?

FORMULATION GROUP. Given P and equal, find generators for GP

Equality oracle

Consider the expression DAG representation of g

3∑

i=1
xiyi − log(x3/y3)

List of expressions ≡
expression DAG sharing

variable leaf nodes

−

+ log

÷×× ×

x1 x2 x3y1 y2 y3

Every function g : Rn → Rm is represented by a DAG whose leaf

nodes are variables and constants and whose intermediate nodes

are mathematical operators

equal(g(x), σg(xπ)) =true if and only if the DAGs rep-
resenting g(x) and σg(xπ) are isomorphic

Reduces the FORMULATION SYMMETRY problem to the
GRAPH ISOMORPHISM problem

GRAPH ISOMORPHISM

Citation: Babai, Automorphism groups, ismorphism,

reconstruction, in Graham, Grötschel, Lovász (eds.),
Handbook of Combinatorics, vol. 2

GI is in NP

It is unknown whether it is in P or NP-complete

Solving GI on rooted DAGs is as hard as solving it on
general graphs

Solving GI on trees has linear complexity

Our DAGs are “close” to trees, can hope they are not
too hard for GI testing

Example

c0 : x6x7 + x8x9 = 1

c1 : x6x8 + x7x9 = 1

x6
6 x7

7 x8
8 x9

9

+0

×2 ×3

+1

×4 ×5

GDAG = group of automorphisms of expression DAG fixing: (a)

root node set having same constr. direction and

coeff. (constraint permutations), (b) operators with same label

and rank and (c) leaf node set (variable permutations)

GDAG = 〈(45)(67)(89), (23)(68)(79), (01)(24)(35)(78)〉
GP is the projection of GDAG to variable indices
〈(6, 7)(8, 9), (6, 8)(7, 9), (7, 8)〉 ∼= D8

Node colors

Let DP = (V,A) be the union of all objective and constraint DAGs in

the MINLP (a.k.a the DAG of P )

Colors on the DAG nodes V are used to identify those subsets of nodes which can

be permuted (e.g. variable and operator nodes can’t be permuted)

1. Root nodes (i.e. constraints) can be permuted if they have the same RHS

2. Operator nodes (including root nodes) can be permuted if they have the

same DAG rank and label; if an operator node is non-commutative, then

the order of the children node must be maintained

3. Constant nodes can be permuted if they have the same DAG rank level

and value

4. Variable nodes can be permuted if they have the same bounds and inte-

grality constraints

The relation (u ∼ v ⇐⇒ u, v have the same color) is an equivalence

relation on V (reflexive, symmetric, transitive)

∼ partitions V into a disjoint union V/ ∼ of equivalence classes V1, . . . , Vp

MINLP formulation groups

Let P be a MINLP and D = (V ,A) be the DAG of P

Let GDAG be the group of automorphisms of D that fix
each color class in V/ ∼
Define φ : GDAG → Sn by φ(π) =projection of π on
variable indices; then
Thm.
φ is a group homomorphism and Imφ ∼= GP

Hence can find GP by computing Imφ

Although the complexity status (P/NP-complete) of the
GRAPH ISOMORPHISM problem is currently unknown,
nauty is a practically efficient software for computing
GDAG

So now we have GP , how do we write “P modulo GP ”?

Symmetry-breaking reformulation

Consider our first example P :

min x1 + x2

3x1 + 2x2 ≥ 1
2x1 + 3x2 ≥ 1

x1, x2 ∈ {0, 1}







P has G(P ) = {(0, 1), (1, 0)}, G∗ = 〈(1, 2)〉 ∼= C2 and
GP = G∗

The orbit GP (0, 1) is the whole of G(P )

We look for a reformulation of P where at least one
representative of each orbit is feasible

Let Q be the reformulation of P consisting of P with the
added constraint x1 ≤ x2

We have G(Q) = {(0, 1)} and G∗ = GQ = 1

Breaking orbital symmetries 1

Every group G ≤ Sn acting on the variable indices N = {1, . . . , n}
partitions N into disjoint orbits (all subsets of N )

This follows from the equiv. rel. i ∼ j ⇔ ∃g ∈ G (g(i) = j)

Let Ω be the set of nontrivial orbits (ω ∈ Ω ⇐⇒ |ω| > 1)

Thm. G acts transitively on each of its orbits

This means that ∀ω ∈ Ω ∀i 6= j ∈ ω ∃g ∈ G (g(i) = j)

Applied to MP, if i, j are distinct variable indices belonging to the same orbit

of GP acting on N , then there is π ∈ GP sending xi to xj

Pick x ∈ G(P ); if P is bounded, for all ω ∈ Ω ∃i ∈ ω s.t. xi is a

component having minimum value over all components of x

By theorem above, ∃π ∈ GP sending xi to xmin ω

Hence x̄ = xπ is s.t. x̄min ω is minimum over all other components of

x̄, and since GP ≤ G∗, x̄ ∈ G(P )

Breaking orbital symmetries 2

Thus, for all ω ∈ Ω there is at least one optimal solution
of P which is feasible w.r.t. the constraints
∀j ∈ ω (xmin ω ≤ xj)

Such constraints are called (orbit-based) symmetry

breaking constraints (SBCs)

Adding these SBCs to P yields a reformulation Q of P
of the narrowing type (prove it!)

Thm. If gω(x) ≤ 0 are SBCs for each orbit ω with “ap-
propriate properties”, then ∀ω ∈ Λ (gω(x) ≤ 0) are also
SBCs

Thus we can combine orbit-based SBCs for
“appropriate properties”

Yields narrowings with fewer symmetric optima

“Appropriate properties”

Notation: g[B](x) ≤ 0 if g(x) only involve variable indices in B

Conditions allowing adjunctions of many SBCs

Thm.

Let ω, θ ⊆ {1, . . . , n} be such that ω ∩ θ = ∅. Consider ρ, σ ∈
GP , and let g[ω](x) ≤ 0 be SBCs w.r.t. ρ,G(P ) and h[θ](x) ≤ 0
be SBCs w.r.t. σ,G(P ). If ρ[ω], σ[θ] ∈ GP [ω∪θ] then the system
of constraints {g[ω](x) ≤ 0, h[θ](x) ≤ 0} is an SBC system for
ρσ.

Breaking the symmetric group

The above SBCs work with any group GP , but their
extent is limited (they may not break all that many
symmetries)

If we find Λ′ ⊆ Λ such that ∀ω ∈ Λ′ the action of GP on ω
is Sym(ω), then there are much tighter SBCs

For all ω ∈ Λ′ let ω− = ω r {max ω} and for all j ∈ ω− let

j+ be the successor of j in ω

The following are valid SBCs:

∀ω ∈ Λ′ ∀j ∈ ω− xj ≤ xj+

which are likely to break many more symmetries

The final attack on the KNP

Decision KNP

Recall the binary KNP variables are used to count the
number of spheres

Suggests simply considering whether a fixed number of

spheres can be placed around a central sphere in a
kissing configuration, or not

This is the decision version of the KNP (dKNP):

Given positive integers n, d, can n unit spheres with disjoint

interior be placed adjacent to a unit sphere centered at the

origin of Rd?

Should eliminate binary variables, yielding a
(nonconvex) NLP, simpler than the original MINLP

In order to find the maximum value for n, we proceed by
bisection on n and solve the dKNP repeatedly

The dKNP formulation

Let N = {1, . . . , n}; the following formulation P correctly
models the dKNP:

max 0

∀i ∈ N
∑

k∈D

x2
ik = 4

∀i ∈ N, j ∈ N : i < j
∑

k∈D

(xik − xjk)
2 ≥ 4

∀i ∈ N, k ∈ D xik ∈ [−2, 2]







If F(P ) 6= ∅ then the answer to the dKNP is YES,
otherwise it is NO

However, solving nonconvex feasibility NLPs is
numerically extremely difficult

Feasibility tolerance

We therefore add a feasibility tolerance variable α:

max α

∀i ∈ N
∑

k∈D

x2
ik = 4

∀i ∈ N, j ∈ N : i < j
∑

k∈D

(xik − xjk)
2 ≥ 4α

∀i ∈ N, k ∈ D xik ∈ [−2, 2]

α ≥ 0







The above formulation Q is always feasible (why?)

Much easier to solve than P , numerically

Q also solves the dKNP: if the optimal α∗ is ≥ 1 then the
answer is YES, otherwise it is NO

INF572 2010/11 – p. 192



The KNP group

The dKNP turns out to have group Sd (i.e. each spatial dimension

can be swapped with any other)

Rewriting the distance constraints as follows:

||xi − xj ||2 =
∑

k∈D

(xik − xjk)2

=
∑

k∈D

(x2
ik + x2

jk + 2xikxjk)

= 2(d +
∑

k∈D

xikxjk)

(for i < j ≤ n) yields an exact reformulation Q′ of Q (prove it)

The formulation group GQ′ turns out to be Sd × Sn (pairs of distinct

spatial dimensions can be swapped, and same for spheres), much

larger than Sd

Yields more effective SBC narrowings

Results

Instance Solver Without SBC With SBC

D N Time Nodes OI Gap Time Nodes OI Gap

2 6 Couenne 4920.16 516000
110150

1 0.04% 100.19 14672 1 0%

2 6 BARON 1200∗ 45259
6015

1 10% 59.63 2785 131 0%

2 7 Couenne 7200† 465500
127220

1 41.8% 7200† 469780
38693

1 17.9%

2 7 BARON 10800 259800
74419

442 83.2% 16632 693162 208 0%

OI: Iteration where optimum was found
†: default Couenne CPU time limit
∗: default BARON CPU time limit

nodes: total nodes
still on tree

Thus, we finally established by MP that k∗(2) = 6

Actually, solutions for k∗(3) and k∗(4) can be found by using MINLP

heuristics (VNS)

The end

